<commit_before>x@{builderDefsPackage
  , autoconf, automake, libtool, doxygen, pkgconfig, bison, flex, unixODBC
  , openssl, openldap, cyrus_sasl, krb5, expat, SDL, libdv, libv4l, alsaLib
  , ...}:
builderDefsPackage
(a :  
let 
  helperArgNames = ["stdenv" "fetchurl" "builderDefsPackage"] ++ 
    [];

  buildInputs = map (n: builtins.getAttr n x)
    (builtins.attrNames (builtins.removeAttrs x helperArgNames));
  sourceInfo = rec {
    baseName="ptlib";
    baseVersion="2.6";
    patchlevel="7";
    version="${baseVersion}.${patchlevel}";
    name="${baseName}-${version}";
    url="mirror://gnome/sources/${baseName}/${baseVersion}/${name}.tar.bz2";
    hash="0zxrygl2ivbciqf97yd9n67ch9vd9gp236w96i6ia8fxzqjq5lkx";
  };
in
rec {
  src = a.fetchurl {
    url = sourceInfo.url;
    sha256 = sourceInfo.hash;
  };

  inherit (sourceInfo) name version;
  inherit buildInputs;

  /* doConfigure should be removed if not needed */
  phaseNames = ["doConfigure" "doMakeInstall"];
      
  meta = {
    description = "Portable Tools from OPAL VoIP";
    maintainers = [ a.lib.maintainers.raskin ];
    platforms = a.lib.platforms.linux;
    broken = true;
  };
  passthru = {
    updateInfo = {
      downloadPage = "http://ftp.gnome.org/sources/ptlib/";
    };
  };
}) x
<commit_msg>ptlib: bump version and fix build for ZHF<commit_after>{ stdenv, fetchurl, pkgconfig, bison, flex, unixODBC
, openssl, openldap, cyrus_sasl, krb5, expat, SDL, libdv, libv4l, alsaLib }:

stdenv.mkDerivation rec {
  name = "ptlib-2.10.10";

  src = fetchurl {
    url = "mirror://gnome/sources/ptlib/2.10/${name}.tar.xz";
    sha256 = "7fcaabe194cbd3bc0b370b951dffd19cfe7ea0298bfff6aecee948e97f3207e4";
  };

  buildInputs = [ pkgconfig bison flex unixODBC openssl openldap 
                  cyrus_sasl krb5 expat SDL libdv libv4l alsaLib ];

  enableParallelBuilding = true;

  patches = [ ./bison.patch ];
      
  meta = with stdenv.lib; {
    description = "Portable Tools from OPAL VoIP";
    maintainers = [ maintainers.raskin ];
    platforms = platforms.linux;
  };

  passthru = {
    updateInfo = {
      downloadPage = "http://ftp.gnome.org/sources/ptlib/";
    };
  };
}

{'fraction_non_alphanumeric': 0.13850415512465375, 'fraction_numerical': 0.03277931671283472, 'mean_word_length': 3.38663967611336, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6710101', 'n_tokens_mistral': 865, 'n_tokens_neox': 814, 'n_words': 154}